

Nr.	Titel	PGM-Nr.
	Bahnbewegungen kartesisch	
1	Bohrungen	151
2	Viereck	152
3	Ecken runden / fasen	153
4	Ecken runden	154
5	Geradenbewegungen	250
6	Kreisbewegungen	251
7	Kreisbogen mit CC, C	206
8	Tangentialer Konturanschluß	207
9	Kreisbögen	208
10	Kreisbogen mit CR	209
	Bahnbewegungen polar	
11	Sechseck polar	213
12	Kreis polar CP	211
13	Kreis tangential-polar CTP	212
14	Polarkoordinaten (allgemein)	252
	Bearbeitungszyklen	
15	Bohren mit Zyklus	201
16	Bohren mit 200er-Zyklen	260
17	Nutenplatte	210
18	Aufnahmeplatte	262
19	Matrize I	265
20	Regelmäßige Lochfläche	220
21	Punktemuster	221
22	Führungsplatte	261
23	Matrize II	266
	Programmteil-Wiederholungen	
24	Sechseck	268
25	Lochplatte - schräge Reihen	270





Nr.	Titel	PGM-Nr.
	Freie Kontur-Programmierung	
26	FK Nocke	288
27	Malteserkreuz	275
28	FK Hammer	289
29	FK Hackenschlüssel	295
	Unterprogramm-Technik	
30	Unterprogramm (Bohrungsgruppen)	215
31	Lochkreissegmente mit mehreren Werkzeugen	280
32	Fräsen mit mehreren Zustellungen	223
	Konturzyklen	
33	Konturzyklen SL 2, Niere ausräumen	240
34	Konturzyklen SL 2, Niere als Insel	241
35	FK Micky Maus SL 2	290
36	FK-SL-Kombination	273
37	DEMO-Baum	276
	Koordinaten-Umrechnungen	
38	Nullpunkt-Verschiebung und Spiegeln	229
39	Koordinaten-Umrechnungen (kombiniert)	232
40	Maßfaktor - Hammer	284
41	Maßfaktor	234
42	Bearbeitungsebene schwenken	S285
43	Bearbeitungsebene schwenken	S286



Aufgabe: **Bohrungen**

Programm(e): _____



Schema: **Bohrungen**

Programm erstellen

Rohteil definieren

Werkzeug definieren

Werkzeug aktivieren

Auf sichere Höhe fahren

Auf Startpunkt fahren

Auf Sicherheitsabstand fahren

Bohren

Zurückziehen des Bohrers

Nächste Position anfahren















Bohren

Zurückziehen des Bohrers


Nächste Bohrung

Bohren

Zurück auf sichere Höhe
Programm Ende

	<i>BEGINN PGM ... MM</i>
	<i>BLK-FORM 0.1 ... X... Y... Z...</i> <i>BLK-FORM 0.2 X... Y... Z...</i>
	<i>TOOL DEF ... L... R...</i>
	<i>TOOL CALL ... S...</i>
	<i>L Z+100 R0 FMAX</i>
	<i>L X... Y... R0 FMAX M3</i>
	<i>L Z+... R FMAX M8</i>
	<i>L Z-... R F100 M</i>
	<i>L Z+2 R0 FMAX</i>
	<i>L X... Y... FMAX</i>
	<i>L Z-...</i>
	<i>L Z+... FMAX</i>
	<i>L X... Y... R0 FMAX</i>
	<i>L Z-22 R0</i>
	<i>L Z+100 R0 FMAX M2 (M30)</i>






Gesamtprogramm

```

0 BEGIN PGM 151 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... ROHTEIL-DEFINITIONEN
3 TOOL DEF 1 L+0 R+4 ..... WERKZEUG-DEFINITION
4 TOOL CALL 1 Z S4000 ..... WERKZEUG-AUFRUF
5 L Z+100 R0 F MAX ..... SICHERE HOEHE
6 L X+20 Y+30 R0 F MAX M3 ..... 1.BOHRUNG
7 L Z+2 R0 F MAX M8
8 L Z-22 R0 F400 ..... BOHREN
9 L Z+2 R0 F MAX
10 L X+50 Y+70 R0 F MAX ..... 2.BOHRUNG
11 L Z-22 R0 F400
12 L Z+2 R0 F MAX
13 L X+75 Y+30 R0 F MAX ..... 3.BOHRUNG
14 L Z-22 R0 F400
15 L Z+100 R0 F MAX M2
16 END PGM 151 MM
    
```



Programm erstellen
 Rohteil definieren

Werkzeug definieren
 Werkzeug aktivieren
 Auf sichere Höhe fahren

Hilfspunkt anfahren R0

Tiefenzustellung

Kontur-Startpunkt RL/RR

Kontur-Koordinaten RL/RR

⋮

Letzter Konturpunkt RL/RR

Hilfspunkt anfahren R0

Freifahren, PGM-Ende

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL DEF... L... R...
TOOL CALL... S...
L Z+... R0 F...

L X... Y... R... F...

L Z... R... F... M...

L X... Y... RR/RL


L X... Y... RR/RL

⋮

L X... Y... RR/RL

L X... Y... R...

L Z... R... F... M...
    
```




Gesamtprogramm


```

0 BEGIN PGM 152 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... ROHTEIL-DEFINITIONEN
3 TOOL DEF 1 L+0 R+8 ..... WERKZEUG-DEFINITIVEN
4 TOOL CALL 1 Z S4000 ..... WERKZEUG-AUFRUF
5 L Z+100 R0 F MAX ..... SICHERE HOEHE
6 L X-30 Y+50 R0 F MAX ..... HILFSPUNKT (R0)
7 L Z-5 R0 F MAX M3 ..... TIEFENZUSTELLUNG
8 L X+0 Y+50 RL F400 ..... STARTPUNKT-KONTUR (RL/RR)
9 L X+50 Y+100
10 L X+100 Y+50
11 L X+50 Y+0
12 L X+0 Y+50 RL ..... LETZTER KONTUR-PUNKT
13 L X-30 R0 F MAX M5 ..... HILFSPUNKT
14 L Z+100 R0 F MAX M2 ..... FREIFAHREN/PGM-ENDE
15 END PGM 152 MM
    
```

Aufgabe: **Ecken runden / fasen**

Programm(e): _____






Gesamtprogramm

```


0 BEGIN PGM 153 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... ROHTEIL-DEFINITION
3 TOOL DEF 1 L+0 R+8 ..... WERKZEUG-DEFINITION
4 TOOL CALL 1 Z S4000 ..... WERKZEUG-AUFRUF
5 L Z+100 R0 F MAX ..... SICHERE HOEHE
6 L X-30 Y+50 R0 F MAX ..... HILFSPUNKT (R0)
7 L Z-5 R0 F MAX M3
8 L X+0 Y+50 RL F200 ..... KONTUR-STARTPUNKT RL
9 L X+50 Y+100
10 RND R10 ..... ECKE RUNDEN
11 L X+100 Y+50
12 L X+50 Y+0
13 CHF 5 ..... FASE
14 L X+0 Y+50 RL
15 L X-30 R0 M5 ..... HILFSPUNKT (RO)
16 L Z+100 R0 F MAX M2 ..... PGM-ENDE
17 END PGM 153 MM
    
```



- Programm erstellen
- Rohteil definieren
- Werkzeug definieren
- Werkzeug aktivieren
- Auf sichere Höhe fahren
- Hilfspunkt anfahren
- Tiefenzustellung
- Weich anfahren
- Kontur
- Weich wegfahren
- Freifahren, PGM-Ende

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL DEF... L... R...
TOOL CALL... S...
L Z...
L X... Y...
L Z...
APPR...
L...
DEP...
L Z...
    
```



Gesamtprogramm

```



0 BEGIN PGM 154 MM
1 BLK FORM 0.1 Z X-20 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+8
4 TOOL CALL 1 Z S4000
5 L Z+100 R0 F MAX
6 L X-30 Y+70 R0 F MAX ..... HILFSPUNKT (R0)
7 L Z-5 R0 F MAX M3
8 APPR LCT X+10 Y+70 R5 RL F400 ..... WEICH ANFAHREN AUF
STARTPUNKT


9 L X+10 Y+90
10 RND R10
11 L X+50 Y+90
12 L Y+50 X+90
13 L X+90 Y+10
14 RND R10
15 L X+50 Y+10
16 L X+10 Y+50
17 L Y+70 ..... LETZTER KONTURPUNKT RL
18 DEP LCT X-30 Y+70 R5 ..... WEICH WEGFAHREN AUF
HILFSPUNKT

19 L Z+100 R0 F MAX M2
20 END PGM 154 MM
    
```

Aufgabe: **Geradenbewegungen**

Programm(e): _____





Gesamtprogramm



```


0 BEGIN PGM 250 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... ROHTEIL-DEFINITION
3 TOOL CALL 7 Z S2500 ..... WERKZEUG-AUFRUF; R4
4 L Z+100 R0 F9999 ..... SICHERE HOEHE
5 L X-20 Y+40 ..... HILFSPUNKT (R0)
6 L Z+2 M3
7 L Z-5 F1000
8 APPR LCT X+5 Y+40 R3 RL F300 ..... 1. KONTURPUNKT
9 L Y+90 (WEICHES ANFAHREN)
10 CHF 20
11 L X+60
12 L X+90 Y+80
13 L Y+40
14 CHF 10
15 L X+60 Y+10
16 L X+5
17 RND R10
18 L Y+40 ..... LETZTER KONTURPUNKT
19 DEP LCT X-20 Y+40 R3 ..... HILFSPUNKT (R0)
20 L Z+100 R0 F MAX M2 ..... FREIFAHREN
21 END PGM 250 MM
    
```



Aufgabe: Kreisbewegungen

Programm(e): _____






Gesamtprogramm

```

0 BEGIN PGM 251 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S2500 ..... R4
4 L Z+100 R0 F9999
5 L X+20 Y-20 ..... HILFSPUNKT (R0)
6 L Z+2 M3
7 L Z-5 F500
8 APPR LCT X+20 Y+30 R3 RL F300 ..... 1. KONTURPUNKT
9 L X+0 (WEICHES ANFAHREN)
10 RND R4
11 L X+15 Y+45
12 CR X+15 Y+60 R+20 DR+
13 L X+0 Y+75
14 CR X+20 Y+95 R+20 DR-
15 L X+40
16 CT X+65 Y+80
17 CC X+75 Y+80
18 C X+85 Y+80 DR+
19 L X+95
20 RND R5
21 L Y+50
22 L X+75 Y+30
23 RND R8
24 L Y+20
25 CC X+60 Y+20
26 C X+45 Y+20 DR-
27 L Y+30
28 RND R9
29 L X+20 ..... LETZTER KONTURPUNKT
30 DEP LCT X+20 Y-20 R3 F500 ..... HILFSPUNKT (R0)
31 L Z+100 R0 F MAX M2
32 END PGM 251 MM
    
```



Programm erstellen
 Rohteil definieren

Werkzeug aktivieren
 Auf sichere Höhe fahren

Hilfspunkt anfahren R0

Tiefenzustellung

Weich anfahren an Kontur

Mittelpunkt setzen

Kreisbewegung


Weich wegfahren (Hilfspkt.)

Freifahren, PGM-Ende

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL... S...
L...
L... R0
L...
APPR... RL/RR
CC...
C...
DEP...
L...
    
```







Gesamtprogramm


```

0 BEGIN PGM 206 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y+50 R0 F MAX ..... HILFSPUNKT
6 L Z-5 R0 F MAX M3
7 APPR LT X+0 Y+20 LEN10 RL F250 M8
8 L X+23,542 RL
9 CC X+50 Y+50 ..... KREISMITTELPUNKT
10 C Y+80 X+23,542 DR+ ..... KREISBEWEGUNG
11 L X+0 RL
12 DEP LT LEN10 ..... WEICH WEGFAHREN (HILFSPKT.)
13 L Z+100 R0 F MAX M2
14 END PGM 206 MM
    
```

Aufgabe: **Tangentialer Konturanschluß
(kartesisch)**

Programm(e): _____






Gesamtprogramm


```

0 BEGIN PGM 207 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 11 Z S2500 ..... R10
4 L Z+100 R0 F MAX ..... SICHERE HOEHE
5 L X+45 Y-25 R0 F MAX ..... HILFSPUNKT
6 L Z-5 F MAX M13
7 APPR LT X+20 Y+0 LEN5 RR F250 ..... WEICH ANFAHREN/
8 L Y+30 ..... STARTPUNKT-KONTUR
9 CT X+70 Y+60 ..... KREISBAHN TANGENTIAL
10 L X+100
11 DEP LT LEN5
12 L Z+100 R0 F MAX M2
13 END PGM 207 MM
    
```



Punkt	X	Y
P ₁	6,645	35,495
P ₂	55,505	69,488

Punkt	X	Y
P ₃	58,995	30,025
P ₄	19,732	21,191




Gesamtprogramm

```

0 BEGIN PGM 208 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX ..... SICHERE HOEHE
5 L X-30 Y+30 R0 F MAX M3 ..... HILFSPUNKT
6 L Z-5 F MAX ..... TIEFE
7 APPR LCT X+5 Y+30 R5 RL F250 M8 ..... WEICH ANFAHREN
8 CC X+15 Y+30 ..... KREISMITTELPUNKT
9 C X+6,645 Y+35,495 DR- ..... KREIS
10 CT X+55,505 Y+69,488 ..... TANGENTIALE KREISBAHN
11 CC X+60 Y+50
12 C X+58,995 Y+30,025 DR-
13 CT X+19,732 Y+21,191
14 CC X+15 Y+30
15 C X+5 Y+30 DR-
16 DEP LCT X-30 Y+30 R5 ..... WEICH WEGFAHREN
17 L Z+100 R0 F MAX M2
18 END PGM 208 MM
    
```






- Programm erstellen
- Rohteil definieren
- Werkzeug aktivieren
- Auf sichere Höhe fahren
- Hilfspunkt anfahren
- Tiefenzustellung
- Weich anfahren
- Kontur
- Weich wegfahren
- Freifahren, PGM-Ende

```


BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2  X... Y... Z...
TOOL CALL... S...
L...
...
...
...
...
...
    
```



Gesamtprogramm

```

0 BEGIN PGM 209 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M13 ..... HILFSPUNKT
6 L Z-5 F MAX
7 APPR LT X+10 Y+0 LEN10 RL F250 ..... WEICH ANFAHREN STARTPKT.
8 L Y+50
9 CR X+90 Y+50 R+80 DR- ..... KREISBOGEN
10 L Y+0
11 DEP LT LEN10 ..... WEICH WEGFAHREN (HILFSPKT.)
12 L Z+100 R0 F MAX M2
13 END PGM 209 MM
    
```



- Programm erstellen
- Rohteil definieren
- Werkzeug aktivieren
- Auf sichere Höhe fahren
- Pol festlegen
- Hilfspunkt mit Polarkoordinaten
- Tiefenzustellung
- Weich anfahren
- Kontur
- ⋮
- Weich wegfahren
- Freifahren, PGM-Ende

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2  X... Y... Z...
TOOL CALL..... S...
L Z...


CC X... Y...
LP PR... PA...

L Z...

APPR PLCT PR... PA... R...

LP PR... PA...
⋮
DEP PLCT PR... PA... R...


L Z...
    
```

Gesamtprogramm

```


0 BEGIN PGM 213 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+0 Y+0 ..... POL
6 LP PR+80 PA+0 R0 F MAX ..... HILFSPUNKT-POLAR
7 L Z-5 F MAX M3
8 APPR PLCT PR+40 PA+0 R5 RR F250 M8 ..... WEICH ANFAHREN/
9 LP PA+60 ..... STARTPUNKT-KONTUR
10 LP PA+120
11 LP PA+180
12 LP PA+240
13 LP PA+300
14 LP PA+360
15 DEP PLCT PR+80 PA+0 R5 ..... WEICH WEGFAHREN
16 L Z+100 R0 F MAX M2
17 END PGM 213 MM
    
```



- Programm erstellen
- Rohteil definieren
- Werkzeug aktivieren
- Auf sichere Höhe fahren
- Pol festlegen
- Hilfspunkt mit Polarkoordinaten
- Kontur
- Hilfspunkt
- Freifahren, PGM-Ende

```

BEGIN PGM ... MM
BLK-FORM 0.1 ... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL ... S...
L ...
...
...
...
...
    
```




Gesamtprogramm

```

0 BEGIN PGM 211 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+50 Y+50 ..... POL
6 LP PR+80 PA+180 R0 F MAX M3 ..... HILFSPUNKT
7 L Z-5 F MAX
8 APPR PLCT PR+40 PA+180 R5 RL F250 M8 ..... WEICH ANFAHREN/
STARTPUNKT-KONTUR
9 CP IPA+360 DR- ..... KREISBAHN POLAR
10 DEP PLCT PR+80 PA+180 R5 ..... WEICH WEGFAHREN
11 L Z+100 F MAX M2
12 END PGM 211 MM
    
```






- Programm erstellen
- Rohteil definieren
- Werkzeug aktivieren
- Auf sichere Höhe fahren
- Pol festlegen
- Hilfspunkt mit Polarkoordinaten
- Kontur
- Hilfspunkt
- Freifahren, PGM-Ende

```


BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL... S...
L...
...
...
...
...
...
    
```




Gesamtprogramm

```

0 BEGIN PGM 212 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+70 Z+0
3 TOOL CALL 13 Z S4000 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M3 ..... HILFSPUNKT
6 L Z-5 F MAX
7 APPR LT X+0 Y+0 LEN5 RL F250 M8 ..... WEICH ANFAHREN/
8 L X+0 Y+30 ..... STARTPUNKT-KONTUR
9 L X+40 Y+60
10 CC Y+20 X+20 ..... POL
11 CTP PR+76,158 PA+23,199 ..... KREISBAHN TANGENTIAL POLAR
12 L Y+0
13 DEP LT LEN5 ..... WEICH WEGFAHREN
14 L Z+100 R0 F MAX M2
15 END PGM 212 MM
    
```






Gesamtprogramm

```

0 BEGIN PGM 252 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X+130 Y+50 F MAX M3
6 L Z-5 F MAX
7 APPR LCT X+92 Y+50 R5 RL F250 M8 ..... WEICH ANFAHREN/
8 L Y+40 ..... STARTPUNKT-KONTUR
9 CC X+92 Y+40 ..... POL
10 LP PR+45 PA-135 ..... LINERAR-POLAR
11 CC X+92 Y+50
12 LP PR+90 PA-150
13 CP PA+148 DR- ..... ZIRKULAR-POLAR
14 L IX+30 IY+0
15 CC ..... POL-POSITIONSUEBERNAHME
16 LP PR+21 PA-60
17 CC X+92 Y+88
18 CTP PR+29 PA-90 ..... TANGENTIAL-POLAR
19 L Y+50
20 DEP LCT X+130 Y+50 R5 ..... WEICH WEGFAHREN
21 L Z+100 R0 F MAX M2
22 END PGM 252 MM
    
```



- Programm erstellen
- Rohteil definieren


- Werkzeug aktivieren
- Zyklus definieren
- Auf sichere Höhe fahren

- Startpunkt 1. Bohrung / Zyklus rufen
- 2. Bohrung / Zyklus rufen
- 3. Bohrung / Zyklus rufen
- Freifahren, PGM-Ende

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL... S...
CYCL DEF...
L Z+...
L X... Y... M99
  



L X... Y... M99
L X... Y... M99
...
    
```

Gesamtprogramm

```

0 BEGIN PGM 201 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1000 ..... R4
4 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-23 ..... TIEFE
  Q206=150 ..... F TIEFENZUSTELLUNG
  Q202=5 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
5 L Z+100 R0 F9999 M3
6 L X+20 Y+30 M99 ..... STARTPOSITION
7 L X+50 Y+70 M99 ..... 2. BOHRUNG
8 L X+75 Y+30 M99 ..... 3. BOHRUNG
9 L Z+100 R0 F MAX M2
10 END PGM 201 MM
    
```



Arbeitsfolge:

- Zentrieren
- Bohren Durchmesser 28 mm
- Ausdrehen
- Reiben
- Gewindebohren


Zyklus 1

Zyklus 203

Zyklus 202

Zyklus 201

Zyklus 2



Gesamtprogramm

```

0 BEGIN PGM 260 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-70
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 9 Z S1000 ..... R6
4 L Z+100 R0 F MAX
5 CYCL DEF 1.0 TIEFBOHREN
6 CYCL DEF 1.1 ABST 2
7 CYCL DEF 1.2 TIEFE -5
8 CYCL DEF 1.3 ZUSTLG 5
9 CYCL DEF 1.4 V.ZEIT 0
10 CYCL DEF 1.5 F200
11 L X+0 Y+0 R0 F9999 M3
12 L Z+2 M99
13 L Z+100 M6
    
```

Zentrieren

Bohren

```



14 TOOL CALL 11 Z S350 ..... R10
15 CYCL DEF 203 UNIVERSAL-BOHREN
    Q200=5 ..... SICHERHEITS-ABSTAND
    Q201=-80 ..... TIEFE
    Q206=100 ..... F TIEFENZUSTELLUNG
    Q202=15 ..... ZUSTELL-TIEFE
    Q210=0 ..... VERWEILZEIT OBEN
    Q203=+0 ..... KOORDINATE OBERFLAECHE
    Q204=20 ..... 2. SICHERHEITS-ABSTAND
    Q212=2 ..... ABNAHMEBETRAG
    Q213=2 ..... SPANBRUECHE
    Q205=5 ..... MINIMALE ZUSTELL-TIEFE
    Q211=0 ..... VERWEILZEIT UNTEN
    Q208=500 ..... F RUECKZUG
16 CYCL CALL M3
17 L Z+100 M6
    
```




Ausdrehen	18	TOOL CALL 13 Z S500	R20	
	19	CYCL DEF 202 AUSDREHEN		
		Q200=2	SICHERHEITS-ABSTAND	
		Q201=-35	TIEFE	
		Q206=250	F TIEFENZUSTELLUNG	
		Q211=0	VERWEILZEIT UNTEN	
		Q208=500	F RUECKZUG	
		Q203=+0	KOORDINATE OBERFLAECHE	
		Q204=20	2. SICHERHEITS-ABSTAND	
		Q214=0	FREIFAHR-RICHTUNG	
		20	CYCL CALL M3	
	21	L Z+100 M6		
Reiben	22	TOOL CALL 13 Z S100	R20	
	23	CYCL DEF 201 REIBEN		
		Q200=5	SICHERHEITS-ABSTAND	
		Q201=-25	TIEFE	
		Q206=100	F TIEFENZUSTUSTELLUNG	
		Q211=0	VERWEILZEIT UNTEN	
		Q208=300	F RUECKZUG	
		Q203=+0	KOORDINATE -OBERFLAECHE	
		Q204=20	2. SICHERHEITSABSTAND	
		24	CYCL CALL M3	
	25	L Z+100 M6		
Gewindebohren	26	TOOL CALL 12 Z S50	R15	
	27	CYCL DEF 2.0 GEWINDEBOHREN		
	28	CYCL DEF 2.1 ABST 5		
	29	CYCL DEF 2.2 TIEFE -45		
	30	CYCL DEF 2.3 V.ZEIT 0		
	31	CYCL DEF 2.4 F100		
	32	L Z-30 M3		
	33	CYCL CALL		
		34	L Z+100 M2	
		35	END PGM 260 MM	

Aufgabe: **Nutenplatte**

Programm(e): _____






Gesamtprogramm

```

0 BEGIN PGM 210 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S1000 ..... R5
4 L Z+100 R0 F9999 M3
5 CYCL DEF 210 NUT PENDELND
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-10 ..... TIEFE
  Q207=100 ..... F FRAESEN
  Q202=5 ..... ZUSTELL-TIEFE
  Q215=0 ..... BEARBEITUNGS-UMFANG
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
  Q216=+50 ..... MITTE 1. ACHSE
  Q217=+25 ..... MITTE 2. ACHSE
  Q218=75 ..... 1. SEITEN-LAENGE
  Q219=15 ..... 2. SEITEN-LAENGE
  Q224=+20 ..... DREHLAGE
6 CYCL CALL

7 CYCL DEF 211 RUNDE NUT
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-10 ..... TIEFE
  Q207=100 ..... F FRAESEN
  Q202=5 ..... ZUSTELL-TIEFE
  Q215=0 ..... BEARBEITUNGS-UMFANG
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
  Q216=+40 ..... MITTE 1. ACHSE
  Q217=+60 ..... MITTE 2. ACHSE
  Q244=50 ..... TEILKREIS-DURCHMESSER
  Q219=12 ..... 2. SEITEN-LAENGE
  Q245=+35 ..... STARTWINKEL
  Q248=180 ..... OEFFNUNGS-WINKEL
8 CYCL CALL

9 L Z+100 M2
10 END PGM 210 MM
    
```



Konventionelle Vorbereitung:

BLK- FORM

**1. Werkzeug
Kreistasche
schruppen**

*TOOL CALL ...
CYCL DEF ...
... M99*

(Aufmaß DR!)

**1. Langloch (groß)
schruppen**

*CYCL DEF ...
CYCL CALL*

**2. Langloch (groß)
schruppen**

*FN 0: Q ...
CYCL CALL
... M6*

Werkzeugwechsel

2. Werkzeug

**1. Langloch (klein)
schlichten**

*TOOL CALL ...
CYCL DEF ...
CYCL CALL*

**2. Langloch (klein)
schlichten**

*FN 0: Q ...
CYCL CALL*

**1. Langloch (groß)
schlichten**

*CYCL DEF ...
CYCL CALL*

**2. Langloch (groß)
schlichten**

*FN 0: Q ...
CYCL CALL*

1. Runde Nut

*CYCL DEF ...
CYCL CALL*

2. Runde Nut

*FN 0: Q ...
CYCL CALL*


**Kreistasche
schlichten**

*CYCL DEF ...
CYCL CALL*

Freifahren, Ende

L Z100 M2

Aufnahmeplatte



Gesamtprogramm


```


0 BEGIN PGM 262 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-15
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 DR+0,5 ..... R3
4 L Z+100 R0 F9999
5 L X+50 Y+50 M3
6 CYCL DEF 5.0 KREISTASCHE
7 CYCL DEF 5.1 ABST 2
8 CYCL DEF 5.2 TIEFE -15
9 CYCL DEF 5.3 ZUSTLG 5 F100
10 CYCL DEF 5.4 RADIUS 10
11 CYCL DEF 5.5 F200 DR+
12 L Z+2 M99
13 CYCL DEF 210 NUT PENDELND
    Q200=2 ..... SICHERHEITS-ABSTAND
    Q201=-6 ..... TIEFE
    Q207=200 ..... F FRAESEN
    Q202=6 ..... ZUSTELL-TIEFE
    Q215=0 ..... BEARBEITUNGS-UMFANG
    Q203=+0 ..... KOORDINATE OBERFLAECHE
    Q204=20 ..... 2. SICHERHEITS-ABSTAND
    Q216=+50 ..... MITTE 1. ACHSE
    Q217=+15 ..... MITTE 2. ACHSE
    Q218=81 ..... 1. SEITEN-LAENGE
    Q219=11 ..... 2. SEITEN-LAENGE
    Q224=+0 ..... DREHLAGE
14 CYCL CALL
15 FN 0: Q217 = +85
16 CYCL CALL
17 L Z+100 M6
    
```



18	TOOL CALL 5 Z S2000	R2,5
19	CYCL DEF 210 NUT PENDELND	
	Q200=2	SICHERHEITS-ABSTAND.
	Q201=-9	TIEFE
	Q207=200	F FRAESEN
	Q202=5	ZUSTELL-TIEFE
	Q215=0	BEARBEITUNGS-UMFANG
	Q203=-6	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+15	MITTE 2. ACHSE
	Q218=76,6	1. SEITEN-LAENGE
	Q219=6,6	2. SEITEN-LAENGE
	Q224=+0	DREHLAGE
20	CYCL CALL M3	
21	FN 0: Q217 = +85	
22	CYCL CALL	
23	CYCL DEF 210 NUT PENDELND	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-6,8	TIEFE
	Q207=200	F FRAESEN
	Q202=6,8	ZUSTELL-TIEFE
	Q215=0	BEARBEITUNGS-UMFANG
	Q203=+0	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+15	MITTE 2. ACHSE
	Q218=81	1. SEITEN-LAENGE
	Q219=11	2. SEITEN-LAENGE
	Q224=+0	DREHLAGE
24	CYCL CALL	
25	FN 0: Q217 = +85	
26	CYCL CALL	
27	CYCL DEF 211 RUNDE NUT	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-15	TIEFE
	Q207=200	F FRAESEN
	Q202=5	ZUSTELL-TIEFE
	Q215=0	BEARBEITUNGS-UMFANG
	Q203=+0	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+50	MITTE 2. ACHSE
	Q244=60	TEILKREIS-DURCHMESSER
	Q219=6,6	2. SEITEN-LAENGE
	Q245=-30	STARTWINKEL
	Q248=60	ÖFFNUNGS-WINKEL
28	CYCL CALL	
29	FN 0: Q245 = +150	
30	CYCL CALL	
31	CYCL DEF 214 KREIST. SCHLICH.	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-15	TIEFE
	Q206=250	F TIEFENZUSTELLUNG
	Q202=8	ZUSTELL-TIEFE
	Q207=250	F FRAESEN
	Q203=+0	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+50	MITTE 2. ACHSE
	Q222=19	ROHTEIL-DURCHMESSER
	Q223=20,02	FERTIGTEIL-DURCHMESSER
32	CYCL CALL	
33	L Z+100 R0 F MAX M2	
34	END PGM 262 MM	







Gesamtprogramm

```



0 BEGIN PGM 265 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 9 Z S500 DR+1 ..... R6
4 L Z+100 R0 F9999
5 CC X+50 Y+50
6 LP PR+70 PA+0
7 L Z+2 M3
8 L Z-10 F200
9 CP PA+360 DR-
10 LP PR+60
11 CP PA+360 DR-
12 LP PR+45 RL
13 CP PA+360 DR-
14 LP PR+65 PA+0 R0
15 L Z+2
16 CYCL DEF 4.0 TASCHENFRAESEN
17 CYCL DEF 4.1 ABST 2
18 CYCL DEF 4.2 TIEFE -5
19 CYCL DEF 4.3 ZUSTLG 5 F100
20 CYCL DEF 4.4 X60
21 CYCL DEF 4.5 Y30
22 CYCL DEF 4.6 F250 DR- RADIUS 7
23 L X+50 Y+50 R0 F MAX M99
24 L Z+100 F MAX M6


```

25	TOOL CALL 8 Z S1000	R5
26	CYCL DEF 212 TASCHE SCHLICH.	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-5	TIEFE
	Q206=250	F TIEFENZUSTELLUNG
	Q202=5	ZUSTELL-TIEFE
	Q207=250	F FRAESEN
	Q203=+0	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+50	MITTE 2. ACHSE
	Q218=60	1. SEITEN-LAENGE
	Q219=30	2. SEITEN-LAENGE
	Q220=5,25	ECKENRADIUS
	Q221=0	AUFMASS
27	CYCL CALL M3	
28	CYCL DEF 215 KREISZ. SCHLICH.	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-10	TIEFE
	Q206=250	F TIEFENZUSTELLUNG
	Q202=10	ZUSTELL-TIEFE
	Q207=250	F FRAESEN
	Q203=+0	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
	Q216=+50	MITTE 1. ACHSE
	Q217=+50	MITTE 2. ACHSE
	Q222=92	ROHTEIL-DURCHMESSER
	Q223=90	FERTIGTEIL-DURCHMESSER
29	CYCL CALL	
30	L Z+100 R0 F MAX M2	
31	END PGM 265 MM	

Aufgabe: **Regelmäßige Lochfläche**

Programm(e): _____





HAUPTPROGRAMM

```



0 BEGIN PGM 220 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-20 ..... TIEFE
  Q206=100 ..... F TIEFENZUSTELLUNG
  Q202=5 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
6 CALL LBL 1


Freifahren, Ende      7 L Z+100 M2
    
```

UP

```

8 LBL 1
9 CYCL DEF 221 MUSTER LINIEN
  Q225=+15 ..... START 1. ACHSE
  Q226=+20 ..... START 2. ACHSE
  Q237=+10 ..... ABSTAND 1. ACHSE
  Q238=+8 ..... ABSTAND 2. ACHSE
  Q242=8 ..... ANZAHL SPALTEN
  Q243=9 ..... ANZAHL ZEILEN
  Q224=+0 ..... DREHLAGE
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
10 LBL 0
11 END PGM 220 MM
    
```





HAUPTPROGRAMM

```

0 BEGIN PGM 221 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-17 ..... TIEFE
  Q206=150 ..... F TIEFENZUSTELLUNG
  Q202=6 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
6 CALL LBL 1
7 CALL LBL 2

Freifahren, Ende      8 L Z+100 M2
    
```

UP

```

9 LBL 1
10 CYCL DEF 220 MUSTER KREIS
  Q216=+90 ..... MITTE 1. ACHSE
  Q217=+10 ..... MITTE 2. ACHSE
  Q244=100 ..... TEILKREIS-DURCHMESSER
  Q245=+90 ..... STARTWINKEL
  Q246=+180 ..... ENDWINKEL
  Q247=15 ..... WINKELSCHRITT
  Q241=7 ..... ANZAHL
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
11 LBL 0
    
```

12 LBL 2



13 CYCL DEF 221 MUSTER LINIEN

Q225=+30.....	START 1. ACHSE
Q226=+30.....	START 2. ACHSE
Q237=+15.....	ABSTAND 1. ACHSE
Q238=+20.....	ABSTAND 2. ACHSE
Q242=5.....	ANZAHL SPALTEN
Q243=2.....	ANZAHL ZEILEN
Q224=+80.....	DREHLAGE
Q200=2.....	SICHERHEITS-ABSTAND
Q203=+0.....	KOORDINATE OBERFLAECHE
Q204=2.....	2. SICHERHEITS-ABSTAND

14 LBL 0


15 END PGM 221 MM





- Arbeitsfolge:**
- Absätze
 - Zentrieren
 - Bohren
 - Gewindebohren

Zyklus 200
Zyklus 200
Zyklus 2



HAUPTPROGRAMM

```

0 BEGIN PGM 261 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-30
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S500 ..... R20
4 L Z+100 R0 F9999
5 L X-50 Y-50
6 L Z+2 M3
7 L Z-10 F200
8 APPR LCT X+30 Y+0 R2 RL
9 L Y+100
10 DEP LCT X+0 Y+150 R2
11 L X+100 F MAX
12 APPR LCT X+70 Y+100 R2 RL
13 L Y+0
14 DEP LCT X+100 Y-50 R2
15 L Z+100 R0 F9999 M6

16 TOOL CALL 4 Z S2000 ..... R2,5
17 CYCL DEF 200 BOHREN
    Q200=2 ..... SICHERHEITS-ABSTAND
    Q201=-3,5 ..... TIEFE
    Q206=250 ..... F TIEFENZUSTELLUNG
    Q202=-3,5 ..... ZUSTELL-TIEFE
    Q210=0 ..... VERWEILZEIT OBEN
    Q203=-10 ..... KOORDINATE OBERFLAECHE
    Q204=20 ..... 2. SICHERHEITS-ABSTAND
18 L Z+5 M3
19 CALL LBL 1
20 L Z+100 M6
    
```




21	TOOL CALL 5 Z S2000	R2,5
22	CYCL DEF 200 BOHREN	
	Q200=2	SICHERHEITS-ABSTAND
	Q201=-15	TIEFE
	Q206=250	F TIEFENZUSTELLUNG
	Q202=5	ZUSTELL-TIEFE
	Q210=0	VERWEILZEIT OBEN
	Q203=-10	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
23	L Z+5 M3	
24	CALL LBL 1	
25	L Z+100 M6	
26	TOOL CALL 6 Z S300	R3
27	CYCL DEF 2.0 GEWINDEBOHREN	
28	CYCL DEF 2.1 ABST 2	
29	CYCL DEF 2.2 TIEFE -12	
30	CYCL DEF 2.3 V.ZEIT 0	
31	CYCL DEF 2.4 F300	
32	L Z+5 M3	
33	CALL LBL 1	
Freifahren, Ende	34 L Z+100 R0 F MAX M2	

UP

35	LBL 1	
36	CYCL DEF 221 MUSTER LINIEN	
	Q225=+15	START 1. ACHSE
	Q226=+10	START 2. ACHSE
	Q237=+70	ABSTAND 1. ACHSE
	Q238=+20	ABSTAND 2. ACHSE
	Q242=2	ANZAHL SPALTEN
	Q243=5	ANZAHL ZEILEN
	Q224=+0	DREHLAGE
	Q200=2	SICHERHEITS-ABSTAND
	Q203=-10	KOORDINATE OBERFLAECHE
	Q204=20	2. SICHERHEITS-ABSTAND
37	LBL 0	
38	END PGM 261 MM	

UP, ENDE




Schema:


Matrize II


Rohteildefinition	<i>BLK FORM</i>	
	<i>CALL PGM...</i>	PGM-Aufruf
Bohren Ø 5 mm	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	UP1-Aufruf
	<i>L Z100... M6</i>	Werkzeugwechsel
Zentrieren, Teilkreis	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	UP2-Aufruf
	<i>L Z100... M6</i>	Werkzeugwechsel
Bohren, Teilkreis	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	UP2-Aufruf
	<i>L Z100... M6</i>	Werkzeugwechsel
Gewindebohren	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	UP2-Aufruf
Freifahren, Ende	<i>L Z... M2</i>	

UP1	<i>LBL 1</i>
Punktemuster, Lochfläche	<i>CYCL DEF...</i>
	<i>LBL 0</i>
UP2	<i>LBL 2</i>
Punktemuster, Kreis	<i>CYCL DEF...</i>
	<i>LBL 0</i>



Hilfsprogramm aus PGM 265 erstellen





HAUPTPROGRAMM

```

0 BEGIN PGM 266 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 CALL PGM 26501 ..... HILFSPROGRAMM
4 TOOL CALL 5 Z S2000 ..... R2,5
5 L Z+100 M3
6 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-6 ..... TIEFE
  Q206=200 ..... F TIEFENZUSTELLUNG
  Q202=6 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=-5 ..... KOORDINATE OBERFLAECHE
  Q204=20 ..... 2. SICHERHEITS-ABSTAND
7 CALL LBL 1
8 L Z+100 R0 F9999 M6

9 TOOL CALL 4 Z S2000 ..... R2
10 L Z+100 M3
11 CYCL DEF 1.0 TIEFBOHREN
12 CYCL DEF 1.1 ABST 2
13 CYCL DEF 1.2 TIEFE -2,8
14 CYCL DEF 1.3 ZUSTLG 2,8
15 CYCL DEF 1.4 V.ZEIT 0
16 CYCL DEF 1.5 F200
17 CALL LBL 2
18 L Z+100 M6

19 TOOL CALL 5 Z S1500 ..... R2,5
20 L Z+100 M3
21 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-15 ..... TIEFE
  Q206=200 ..... F TIEFENZUSTELLUNG
  Q202=8 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=20 ..... 2. SICHERHEITS-ABSTAND
22 CALL LBL 2
23 L Z+100 M6
    
```



24 TOOL CALL 6 Z S300 R3
 25 L Z+100 M3
 26 CYCL DEF 2.0 GEWINDEBOHREN
 27 CYCL DEF 2.1 ABST 2
 28 CYCL DEF 2.2 TIEFE -12
 29 CYCL DEF 2.3 V.ZEIT 1
 30 CYCL DEF 2.4 F300
 31 CALL LBL 2

 32 L Z+100 R0 F MAX M2

UP


33 LBL 1
 34 CYCL DEF 221 MUSTER LINIEN
 Q225=+30 START 1. ACHSE
 Q226=+42 START 2. ACHSE
 Q237=+10 ABSTAND 1. ACHSE
 Q238=+8 ABSTAND 2. ACHSE
 Q242=5 ANZAHL SPALTEN
 Q243=3 ANZAHL ZEILEN
 Q224=+0 DREHLAGE
 Q200=2 SICHERHEITS-ABSTAND
 Q203=-6 KOORDINATE OBERFLAECHE
 Q204=20 2. SICHERHEITS-ABSTAND
 35 LBL 0


 36 LBL 2
 37 CYCL DEF 220 MUSTER KREIS
 Q216=+50 MITTE 1. ACHSE
 Q217=+50 MITTE 2. ACHSE
 Q244=80 TEILKREIS-DURCHMESSER
 Q245=+0 STARTWINKEL
 Q246=+360 ENDWINKEL
 Q247=30 WINKELSCHRITT
 Q241=12 ANZAHL
 Q200=2 SICHERHEITS-ABSTAND
 Q203=+0 KOORDINATE OBERFLAECHE
 Q204=20 2. SICHERHEITS-ABSTAND
 38 LBL 0
 39 END PGM 266 MM



Aufgabe: **Sechseck**

Programm(e): _____





Gesamtprogramm

```



0 BEGIN PGM 268 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 12 Z S500 ..... R15
4 L Z+100 R0 F9999
5 CC X+50 Y+50 ..... POL
6 LP PR+80 PA+0 R0 ..... HILFSPUNKT
7 L Z-5 M3
8 APPR PLCT PR+40 PA+0 R2 RL F100 ..... STARTPUNKT

9 LBL 1 ..... LABEL SETZEN
10 LP PR+40 IPA-60
11 CALL LBL 1 REP 5/5 ..... LABEL-AUFRUF MIT
WIEDERHOLUNGEN


12 DEP PLCT PR+80 PA+0 R2 F200 ..... HILFSPUNKT
13 L Z+100 R0 F MAX M2
14 END PGM 268 MM
    
```

Aufgabe: Lochplatte - schräge Reihen

Programm(e): _____




Mäanderförmiges Abfahren
(Zeilenförmiges Abfahren)



Bohrbild

<i>L X... Y...</i> <i>L Z...</i>	Absolute Vorposition Sicherheitsabstand
<i>LBL 1</i>	Label 1 setzen
<i>L IX... M99</i>	Quer-Schritt zur 1. Bohrung
<i>LBL 2</i>	Label 2 setzen
<i>L IY... M99</i>	Längs-Schritt (Spalte nach oben)
<i>CALL LBL 2 REP...</i>	
<i>L IX... M99</i>	Quer-Schritt zur 2. Spalte
<i>LBL 3</i>	Label 3 setzen
<i>L IY... M99</i>	Längs-Schritt (Spalte nach unten)
<i>CALL LBL 3 REP...</i> <i>CALL LBL 1 REP...</i>	Restliche Gruppen



HAUPTPROGRAMM

```

0 BEGIN PGM 270 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 L Z+100 R0 F MAX

4 TOOL CALL 5 Z S4000 ..... R2,5
5 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-15 ..... TIEFE
  Q206=250 ..... F TIEFENZUSTELLUNG
  Q202=8 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=20 ..... 2. SICHERHEITS-ABSTAND

6 L X-10 Y+10 R0 F9999 M3
7 L Z+2

8 LBL 1
9 L IX+20 M99



10 LBL 2
11 L IX+3 IY+15 M99
12 CALL LBL 2 REP 4/4


13 L IX+10 M99

14 LBL 3
15 L IX-3 IY-15 M99
16 CALL LBL 3 REP 4/4

17 CALL LBL 1 REP 2/2

18 L Z+100 R0 F MAX M2
19 END PGM 270 MM
    
```







Gesamtprogramm

```


0 BEGIN PGM 288 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 10 Z S4000 ..... R7,5
4 L Z+100 R0 F MAX
5 L X-15 Y+35 R0 F MAX M3
6 L Z-5 R0 F MAX
7 APPR LCT X+5 Y+35 R5 RL F400
8 FC DR- R25 CCX+30 CCY+35
9 FLT
10 FCT DR- R15 CCX+50 CCY+75
11 FLT
12 FCT DR- R20 CCX+75 CCY+25
13 FLT
14 FCT X+5 Y+35 DR- R25 CCX+30 CCY+35
15 DEP LCT X-15 Y+35 R5
16 L Z+100 R0 F MAX M2
17 END PGM 288 MM
    
```


Aufgabe: **Malteserkreuz**

Programm(e): _____







HAUPTPROGRAMM

```

0 BEGIN PGM 275 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 8 Z S250074 ..... R5
4 L Z+100 R0 F MAX
5 L Z+2 R0 F MAX M3
6 CALL LBL 10

7 LBL 5
8 CYCL DEF 10.0 DREHUNG
9 CYCL DEF 10.1 IROT-90
10 CALL LBL 10
11 CALL LBL 5 REP 2/2

12 CYCL DEF 10.0 DREHUNG
13 CYCL DEF 10.1 ROT+0



14 L Z+100 R0 F MAX M2
    
```


UP

```

15 LBL 10
16 CC X+0 Y+0
17 LP PR+70 PA+180 R0 F1000
18 L Z-10
19 APPR PLCT PR+50 PA+180 R2 RL F500
20 FPOL X+0 Y+0
21 FC DR- R50 CCX+0 CCY+0
22 FL AN+65 PDX+0 PDY+0 D48
23 FSELECT 4
24 FC DR+ R38 CCPR+75 CCPA+135
25 FSELECT 2
26 FL AN+25 PDX+0 PDY+0 D48
27 FSELECT 1
28 FC PR+50 PA+90 DR- R50 CCA+90
29 DEP PLCT PR+70 PA+90 R2
30 LBL 0
31 END PGM 275 MM
    
```









Gesamtprogramm

```

0 BEGIN PGM 289 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 11 Z S4000
4 L Z+100 R0 F MAX
5 CC X+30 Y+40
6 LP PR+60 PA+30 R0 F MAX M3
7 L Z+2 R0 F MAX
8 L Z-5 R0 F100
9 APPR PLCT PR+37,5 PA+30 R5 RL F400
10 FL LEN10 AN-60
11 FL AN-150
12 FCT DR+ R12,5
13 FLT AN-100
14 FCT DR- R11 CCX+30 CCY+15
15 FLT
16 FCT DR- R11 CCX+15 CCY+65
17 FLT AN-55
18 FCT DR+ R15
19 FLT AN+30
20 FPOL X+30 Y+40
21 FL PR+37,5 PA+30 LEN10 AN-60
22 DEP PLCT PR+60 PA+30 R5
23 L Z+100 R0 F MAX M2
24 END PGM 289 MM

```






Gesamtprogramm

```

0 BEGIN PGM 295 MM
1 BLK FORM 0.1 Z X+0 Y-35 Z-20
2 BLK FORM 0.2 X+160 Y+35 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX
5 L X-10 Y-10 R0 F MAX
6 L Z-5 R0 F MAX M3
7 APPR LCT X+0 Y+0 R2 RR F400
8 FL AN+0
9 FCT DR- R70
10 FCT DR+ R60 CCX+105 CCY+28
11 FSELECT 2
12 FCT DR+ R12
13 FCT DR+ R18 CCX+138 CCY-5
14 FLT LEN5 AN+170
15 FL X+134 AN-150
16 FSELECT 1
17 FL LEN4 IAN-90
18 FCT DR- R8
19 FCT DR- R60 CCX+105
20 FSELECT 1
21 FCT DR- R40 CCX+100
22 FCT Y+20 DR+ R5
23 FSELECT 2
24 FLT X+0 AN+180
25 FSELECT 1
26 L Y+0
27 DEP LCT X-10 Y-10 R2
28 L Z+100 R0 FMAX M2
29 END PGM 295 MM
    
```

Aufgabe: **Unterprogramm
(Bohrungsgruppen)**

Programm(e): _____



Programm erstellen

```

BEGIN PGM... MM
:
:
TOOL CALL...
CYCL DEF...
L Z+...
L X... Y...
L Z...
CALL LBL...
L X... Y...
CALL LBL...
L X... Y...
CALL LBL...

```

Werkzeug aktivieren
Zyklus definieren
Auf sichere Höhe

Startpos. Bohrgrup. ①
Sicherheitsabstand

UP-Aufruf

Startpos. Bohrgrup. ②

UP-Aufruf

Startpos. Bohrgrup. ③
UP-Aufruf

Freifahren, Ende

```
L Z +100 R0 F9999 M2
```

UP

```


LBL...

```

UP-Ende

```
LBL 0
```





HAUPTPROGRAMM

```

0 BEGIN PGM 215 MM
1 BLK FORM 0.1 Z X+0 Y-70 Z-20
2 BLK FORM 0.2 X+100 Y+0 Z+0
3 TOOL CALL 4 Z S4000 ..... WERKZEUG-AUFRUF, R2
4 CYCL DEF 200 BOHREN
  Q200=2 ..... SICHERHEITS-ABSTAND
  Q201=-18 ..... TIEFE
  Q206=200 ..... F TIEFENZUSTELLUNG
  Q202=4 ..... ZUSTELL-TIEFE
  Q210=0 ..... VERWEILZEIT OBEN
  Q203=+0 ..... KOORDINATE OBERFLAECHE
  Q204=2 ..... 2. SICHERHEITS-ABSTAND
5 L Z+100 R0 F MAX ..... SICHERE HOEHE
6 L X+20 Y-10 R0 F MAX ..... STARTPUNKT-BOHRGRUPPE ①
7 L Z+2 R0 F MAX M13
8 CALL LBL 1 ..... UP-AUFRUF

9 L X+63 Y-12 F MAX ..... STARTPOSITION-BOHRGRUPPE ②
10 CALL LBL 1 ..... UP-AUFRUF

11 L X+42 Y-32 F MAX ..... STARTPOSITION-BOHRGRUPPE ③
12 CALL LBL 1 ..... UP-AUFRUF

Freifahren, Ende 13 L Z+100 R0 F MAX M2
    
```

UP

```


14 LBL 1
15 CYCL CALL
16 L IX+15 F MAX M99
17 L IY-15 F MAX M99 ..... BOHRBILD
18 L IX-15 F MAX M99
19 LBL 0
20 END PGM 215 MM
    
```

UP, ENDE



Aufgabe: Lochkreissegmente mit mehreren Werkzeugen

Programm(e): _____



Schema:

Verschachtelung bei doppelten Lochkreissegmenten

Konventionelle Vorbereitung:

BLK-FORM

Zentrieren

*TOOL CALL ...
CYCL DEF / L Z+100*

CALL LBL 1

Bohren

*TOOL CALL ...
CYCL DEF ...*

CALL LBL 1

Gewindebohren

*TOOL CALL ...
CYCL DEF ...*

CALL LBL 1

Freifahren, Ende

L Z100 M2

UP1

LBL 1

CC X... Y...

CALL LBL 2

CC X... Y...

CALL LBL 2

⋮

UP1-Ende

LBL 0

**UP2,
Lochkreissegm.**

LBL 2

*LP PR ... PA ... M3
L Z+2 M99*

LBL 3

⋮

CALL LBL 3 REP ...

Teilwiederholungen

LP PR ... PA ...

LBL 4

⋮

CALL LBL 4 REP ...

UP2-Ende

LBL 0

Mittelpunkt-Links

Lochkreissegm. rufen

Mittelpunkt-Rechts


Lochkreissegm. rufen

Startposition

Restliche Bohrungen



HEIDENHAIN



HAUPTPROGRAMM

```

0 BEGIN PGM 280 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+150 Y+60 Z+0
3 TOOL CALL 4 Z S2000 ..... R2
4 L Z+100 R0 F9999
5 CYCL DEF 1.0 TIEFBOHREN
6 CYCL DEF 1.1 ABST 2
7 CYCL DEF 1.2 TIEFE -3,5
8 CYCL DEF 1.3 ZUSTLG 3,5
9 CYCL DEF 1.4 V.ZEIT 0
10 CYCL DEF 1.5 F200
11 CALL LBL 1
12 L Z+100 M6

13 TOOL CALL 5 Z S1500 ..... R2,5
14 CYCL DEF 200 BOHREN
    Q200=2 ..... SICHERHEITS-ABSTAND
    Q201=-15 ..... TIEFE
    Q206=250 ..... F TIEFENZUSTELLUNG
    Q202=8 ..... ZUSTELL-TIEFE
    Q210=0 ..... VERWEILZEIT OBEN
    Q203=+0 ..... KOORDINATE OBERFLAECHE
    Q204=20 ..... 2. SICHERHEITS-ABSTAND
15 CALL LBL 1
16 L Z+100 M6

17 TOOL CALL 6 Z S300 ..... R3
18 CYCL DEF 2.0 GEWINDEBOHREN
19 CYCL DEF 2.1 ABST 2
20 CYCL DEF 2.2 TIEFE -12
21 CYCL DEF 2.3 V.ZEIT 1
22 CYCL DEF 2.4 F300
23 CALL LBL 1

24 L Z+100 R0 M2
    
```



UP

25 LBL 1
26 CC X+40 Y+0
27 CALL LBL 2
28 CC X+110 Y+0
29 CALL LBL 2
30 LBL 0

31 LBL 2
32 LP PR+55 PA+120 R0 M3
33 L Z+2 M99

34 LBL 3
35 LP IPR-10 M99
36 CALL LBL 3 REP 3/3



37 LP PR+15 PA+60 R0 M99


38 LBL 4
39 LP IPR+10 M99
40 CALL LBL 4 REP 3/3

41 LBL 0
42 END PGM 280 MM

Aufgabe: Fräsen mit mehreren Zustellungen

Programm(e): _____





HAUPTPROGRAMM

```

0 BEGIN PGM 223 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX M3
5 L X-30 Y+70 R0 F MAX ..... STARTPOSITION
6 L Z+0 F MAX

7 LBL 2
8 L IZ-5 R0 F MAX M3 ..... ZUSTELLUNG
9 CALL LBL 1 ..... AUFRUF DER KONTUR

10 CALL LBL 2 REP 5/5 ..... WEITERE KONTURSCHNITTE

11 L Z+100 R0 F MAX M2
    
```

Freifahren, Ende

UP, Kontur



```


12 LBL 1
13 APPR LCT X+10 Y+70 R5 RL F250 M3
14 L X+10 Y+90 RL
15 RND R10
16 L X+50 Y+90
17 RND R20
18 L X+90 Y+50
19 RND R20 ..... KONTUR
20 L X+90 Y+10
21 RND R10
22 L X+50 Y+10
23 RND R20
24 L X+10 Y+50
25 RND R20
26 L X+10 Y+70
27 DEP LCT X-20 Y+70 R5 F500
28 LBL 0
29 END PGM 223 MM
    
```

UP-Ende

**Aufgabe: Konturzyklen SL II,
Niere ausräumen**

Programm(e): _____





HAUPTPROGRAMM

```

0 BEGIN PGM 240 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1500 ..... R4
4 L Z+100 R0 F9999 M3
5 CYCL DEF 14.0 KONTUR
6 CYCL DEF 14.1 KONTURLABEL 1
7 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-18 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+0,5 ..... AUFMASS SEITE
  Q4=+0,5 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATE OBERFLAECHE
  Q6=2 ..... SICHERHEITS-ABSTAND
  Q7=+10 ..... SICHERE HOEHE
  Q8=0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
8 CYCL DEF 22.0 RAEUMEN
  Q10=10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q18=0 ..... VORRAEUM-WERKZEUG
  Q19=150 ..... VORSCHUB PENDELN
9 CYCL CALL
10 L Z+100 R0 F MAX M6

11 TOOL CALL 8 Z S2000 ..... R5
12 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
13 CYCL CALL M3
14 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
15 CYCL CALL M3

16 L Z+100 R0 F MAX M2
  
```

Freifahren, Ende

Lösung:

Konturzyklen SL II, Niere ausräumen



UP


17 LBL 1
18 L X+5 Y+30 RR
19 CC X+15 Y+30
20 C X+6,645 Y+35,495 DR-
21 CT X+55,505 Y+69,488
22 CT X+58,995 Y+30,025
23 CT X+19,732 Y+21,191
24 C X+5 Y+30 DR-
25 LBL 0
26 END PGM 240 MM



**Aufgabe: Konturzyklen SL II,
Niere als Insel**

Programm(e): _____





HAUPTPROGRAMM

```

0 BEGIN PGM 241 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 14.0 KONTUR
6 CYCL DEF 14.1 KONTURLABEL 1 /2
7 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-18 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+0,5 ..... AUFMASS SEITE
  Q4=+0,5 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATE OBERFLAECHE
  Q6=2 ..... SICHERHEITS-ABSTAND
  Q7=+10 ..... SICHERE HOEHE
  Q8=0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
8 CYCL DEF 22.0 RAEUMEN
  Q10=10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q18=0 ..... VORRAEUM-WERKZEUG
  Q19=150 ..... VORSCHUB PENDELN
9 CYCL CALL
10 L Z+100 R0 F MAX M6
11 TOOL CALL 8 Z S2000 ..... R4
12 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
13 CYCL CALL M3
14 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
15 CYCL CALL M3
16 L Z+100 R0 F MAX M2
  
```


Freifahren, Ende


UP

17 LBL 1 INSEL
18 L X+5 Y+30 RL
19 CC X+15 Y+30
20 C X+6,645 Y+35,495 DR-
21 CT X+55,505 Y+69,488
22 CT X+58,995 Y+30,025
23 CT X+19,732 Y+21,191
24 C X+5 Y+30 DR-
25 LBL 0

26 LBL 2 KAMMER
27 L X-10 Y-10 RR
28 L Y+110
29 L X+110
30 L Y-10
31 L X-10
32 LBL 0
33 END PGM 241 MM







HAUPTPROGRAMM

```

0 BEGIN PGM 290 MM
1 ..... MICKY MIT 20ER ZYKLEN
2 ..... B06
3 BLK FORM 0.1 Z X-70 Y-90 Z-12
4 BLK FORM 0.2 X+70 Y+30 Z+0
5 TOOL CALL 7 Z S2500 ..... R4
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 KONTUR
8 CYCL DEF 14.1 KONTURLABEL 1 /2
9 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-10 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+20 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
10 CYCL DEF 22.0 RAEUMEN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q18=0 ..... VORRAEUM-WERKZEUG
  Q19=150 ..... VORSCHUB PENDELN
11 L X-65 Y+0 M99

12 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
13 CYCL CALL
14 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
15 CYCL CALL



Freifahren, Ende
16 L Z+100 R0 F MAX M2
    
```


UP

```
17 LBL 1
18 L X-62 Y+0 RL
19 FC DR- R22 CLSD+ CCX-40 CCY+0
20 FCT DR+ R38 CCX+0
21 FSELECT 1
22 FCT DR- R22 CCX+40 CCY+0
23 FCT DR+ R10
24 FLT AN-60 PDX+0 PDY+0 D0
25 FSELECT 3
26 FCT DR- R8
27 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
28 FLT X-40 AN+180
29 FL AN+90
30 FL AN+30
31 FCT DR+ R9 CCX-17
32 FLT X-30 PAR31 DP18
33 FSELECT 2
34 FL AN+90
35 FL PAR33 DP8
36 FSELECT 2
37 FCT DR+ R9 CCX-17
38 FCT DR- R22 CLSD- CCX-40 CCY+0
39 FSELECT 2
40 LBL 0

41 LBL 2
42 L X-74 Y+20 RR
43 L Y+35
44 L X+74
45 L Y-100
46 L X-74
47 L Y+0
48 LBL 0
49 END PGM 290 MM
```







HAUPTPROGRAMM

```

0 BEGIN PGM 273 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-30
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 CYCL DEF 14.0 KONTUR
5 CYCL DEF 14.1 KONTURLABEL 1 /2
6 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-10 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATE OBERFLAECHE
  Q6=2 ..... SICHERHEITS-ABSTAND
  Q7=+20 ..... SICHERE HOEHE
  Q8=0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
7 CALL LBL 10

8 CYCL DEF 14.0 KONTUR
9 CYCL DEF 14.1 KONTURLABEL 1 /3
10 FN 0: Q5 = -10
11 CALL LBL 10
12 L Z+100 R0 F MAX M6

13 TOOL CALL 4 Z S2000 ..... R2
14 CYCL DEF 14.0 KONTUR
15 CYCL DEF 14.1 KONTURLABEL 1 /3
16 CALL LBL 11

17 CYCL DEF 14.0 KONTUR
18 CYCL DEF 14.1 KONTURLABEL 1 /2
19 FN 0: Q5 = +0
20 CALL LBL 11


Freifahren, Ende
21 L Z+100 R0 F MAX M2
    
```

UP


22	LBL 10	
23	CYCL DEF 22.0 RAEUMEN	
	Q10=10	ZUSTELL-TIEFE
	Q11=100	VORSCHUB TIEFENZUSTELLUNG
	Q12=200	VORSCHUB RAEUMEN
	Q18=0	VORRAEUM-WERKZEUG
	Q19=150	VORSCHUB PENDELN
24	CYCL CALL M3	
25	LBL 0	
26	LBL 11	
27	CYCL DEF 23.0 SCHLICHTEN TIEFE	
	Q11=100	VORSCHUB TIEFENZUSTELLUNG
	Q12=200	VORSCHUB RAEUMEN
28	CYCL CALL M3	
29	CYCL DEF 24.0 SCHLICHTEN SEITE	
	Q9=-1	DREHSINN
	Q10=5	ZUSTELL-TIEFE
	Q11=100	VORSCHUB TIEFENZUSTELLUNG
	Q12=200	VORSCHUB RAEUMEN
	Q14=+0	AUFMASS SEITE
30	CYCL CALL	
31	LBL 0	
32	LBL 1	
33	L X-40 Y+0 RR	
34	L Y+40	
35	RND R5	
36	L X+40	
37	RND R5	
38	L Y-40	
39	RND R5	
40	L X-40	
41	RND R5	
42	L Y+0	
43	LBL 0	
44	LBL 2	
45	CC X+0 Y+0	
46	FPOL X+0 Y+0	
47	LP PR+22 PA+45 RL	
48	FC DR- R4 CLSD+	
49	FLT AN-91,85	
50	FCT DR- R16 CCX+0 CCY+0	
51	FSELECT 1	
52	FLT AN-178,15	
53	FCT DR- R4	
54	FLT AN+88,15	
55	FCT DR- R16 CCX+0 CCY+0	
56	FLT AN+1,85	
57	FCT PR+22 PA+45 DR- R4 CLSD-	
58	FSELECT 2	
59	LBL 0	

60 LBL 3
61 CC X+0 Y+0
62 LP PR+28 PA+180 RL
63 FC Y+14 DR- R28 CCX+0 CCY+0
64 FSELECT 2
65 FL AN+0
66 FCT X-14 DR+ R5
67 FLT AN+90
68 FC X+14 DR- R28 CCX+0 CCY+0
69 FSELECT 1
70 FL AN-90
71 FCT Y+14 DR+ R5
72 FLT AN+0
73 FC Y-14 DR- R28 CCX+0 CCY+0
74 FSELECT 1
75 FL AN-180
76 FCT X+14 DR+ R5
77 FLT AN-90
78 FC X-14 DR- R28 CCX+0 CCY+0
79 FSELECT 2
80 FL AN+90
81 FCT Y-14 DR+ R5
82 FLT AN+180
83 FC X-28 Y+0 DR- R28 CCX+0 CCY+0
84 LBL 0
85 END PGM 273 MM





P0 ≠ Konturpunkt



Punkt	X	Y	Punkt	X	Y
P0	0,0	0,0	M1	-5,5	13,5
P1	-	9,0	M2	2,5	24,0
P2	-	10,0	M3	12,0	13,0
			M4	-	-2,0
			M5	-3,0	-18,0
			M6	8,0	-18,0

Schema:

DEMO-Baum

**Konventionelle
Vorbereitung**

```
BLK FORM  
TOOL CALL 1 . . .
```

SL II-Zyklen

```
CYCL DEF 14.0 KONTUR  
CYCL DEF 14.1 KONTURLABEL 1  
CYCL DEF 20.0 KONTUR-DATEN  
CYCL DEF 22.0 RAEUMEN
```

1. Nullpunkt
Kontur-Aufruf

```
CYCL DEF 7.0 NULLPUNKT  
CYCL CALL . . .
```

2. Nullpunkt
Kontur-Aufruf

•
•
•

Freifahren, Ende

```
L Z100 . . . M2
```

Kontur, UP 1

```
LBL 1
```



•
•
•
•
•
•

```
LBL 0
```

**Nullpunkt-Tabelle
erstellen**

```
BEGIN 276 .D MM  
D X Y Z C A  
0 +0 +0 +0 +0 +0  
1 +20 +30 +0 +0 +0  
2 • • • • •  
3 • • • • •  
4 • • • • •  
• • • • •  
• • • • •  
• • • • •  
• • • • •  
• • • • •  
• • • • •  
END
```





P0 ≠ Konturpunkt

Punkt	X	Y	Punkt	X	Y
P0	0,0	0,0	M1	-5,5	13,5
P1	-	9,0	M2	2,5	24,0
P2	-	10,0	M3	12,0	13,0
			M4	-	-2,0
			M5	-3,0	-18,0
			M6	8,0	-18,0

HAUPTPROGRAMM

```

0 BEGIN PGM 276 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-10
2 BLK FORM 0.2 X+130 Y+170 Z+0
3 TOOL CALL 3 Z S3000 ..... R1,5
4 L Z+100 R0 F MAX
5 L X+0 Y+0 R0 F9999
6 L Z+2 M3
7 CYCL DEF 14.0 KONTUR
8 CYCL DEF 14.1 KONTURLABEL 1
9 CYCL DEF 20.0 KONTUR-DATEN
   Q1=-5 ..... FRAESTIEFE
   Q2=1 ..... BAHN-UEBERLAPPUNG
   Q3=+0 ..... AUFMASS SEITE
   Q4=+0 ..... AUFMASS TIEFE
   Q5=+0 ..... KOORDINATE OBERFLAECHE
    
```



Q6=2.....	SICHERHEITS-ABSTAND
Q7=+5	SICHERE HOEHE
Q8=0.....	RUNDUNGSRADIUS
Q9=-1	DREHSINN
10 CYCL DEF 22.0 RAEUMEN	
Q10=5.....	ZUSTELL-TIEFE
Q11=100.....	VORSCHUB TIEFENZUSTELLUNG
Q12=200.....	VORSCHUB RAEUMEN
Q18=0.....	VORRAEUM-WERKZEUG
Q19=150.....	VORSCHUB PENDELN
11 CYCL DEF 7.0 NULLPUNKT	
12 CYCL DEF 7.1 #1	
13 CYCL CALL	
14 CYCL DEF 7.0 NULLPUNKT	
15 CYCL DEF 7.1 #2	
16 CYCL CALL	
17 CYCL DEF 7.0 NULLPUNKT	
18 CYCL DEF 7.1 #3	
19 CYCL CALL	
20 CYCL DEF 7.0 NULLPUNKT	
21 CYCL DEF 7.1 #4	
22 CYCL CALL	
23 CYCL DEF 7.0 NULLPUNKT	
24 CYCL DEF 7.1 #5	
25 CYCL CALL	
26 CYCL DEF 7.0 NULLPUNKT	
27 CYCL DEF 7.1 #6	
28 CYCL CALL	
29 CYCL DEF 7.0 NULLPUNKT	
30 CYCL DEF 7.1 #7	
31 CYCL CALL	
32 CYCL DEF 7.0 NULLPUNKT	
33 CYCL DEF 7.1 #8	
34 CYCL CALL	
Freifahren, Ende	35 L Z+100 R0 F MAX M2

UP

```

36 LBL 1
37 L X+2,5 Y+32,5 RR
38 FC DR- R8,5 CCX+2,5 CCY+24
39 FC Y+10 DR- R7,5 CCX+12 CCY+13
40 FSELECT 2
41 FC DR- R40 CCY-2
42 FSELECT 2
43 FCT DR+ R10
44 FCT DR- R2 CCX+8 CCY-18
45 FSELECT 1
46 FCT DR+ R8
47 FCT DR- R2 CCX-3 CCY-18
48 FSELECT 2
49 FCT Y+9 DR+ R36
50 FC DR- R6,5 CCX-5,5 CCY+13,5
51 FC X+2,5 Y+32,5 DR- R8,5 CCX+2,5 CCY+24
52 FSELECT 2
53 LBL 0
54 END PGM 276 MM

```




Nullpunkt-Tabelle

BEGIN 276		.D MM				
D	X	Y	Z	C	A	
0	+0	+0	+0	+0	+0	
1	+20	+30	+0	+0	+0	
2	+60	+30	+0	+0	+0	
3	+100	+30	+0	+0	+0	
4	+40	+80	+0	+0	+0	
5	+80	+80	+0	+0	+0	
6	+20	+130	+0	+0	+0	
7	+60	+130	+0	+0	+0	
8	+100	+130	+0	+0	+0	
9	+0	+0	+0	+0	+0	
10	+0	+0	+0	+0	+0	
11	+0	+0	+0	+0	+0	
12	+0	+0	+0	+0	+0	
13	+0	+0	+0	+0	+0	
14	+0	+0	+0	+0	+0	
15	+0	+0	+0	+0	+0	
16	+0	+0	+0	+0	+0	
17	+0	+0	+0	+0	+0	
18	+0	+0	+0	+0	+0	
19	+0	+0	+0	+0	+0	
20	+0	+0	+0	+0	+0	

[END]





HAUPTPROGRAMM

```

0 BEGIN PGM 229 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S4000 ..... R4
4 L Z+100 R0 F MAX
5 CYCL DEF 7.0 NULLPUNKT
6 CYCL DEF 7.1 X+50
7 CYCL DEF 7.2 Y+50
8 CALL LBL 1

9 CYCL DEF 8.0 SPIEGELN
10 CYCL DEF 8.1 X
11 CALL LBL 1

12 CYCL DEF 8.0 SPIEGELN
13 CYCL DEF 8.1 Y
14 CALL LBL 1

15 CYCL DEF 8.0 SPIEGELN
16 CYCL DEF 8.1 X Y
17 CALL LBL 1

18 L Z+100 R0 F MAX M2
    
```

Freifahren, Ende



UP, Kontur


```

19 LBL 1
20 L X+0 Y+0 R0 F MAX M3 ..... HILSPUNKT R0
21 L Z+2 F MAX
22 L Z-15 R0 F100 ..... TIEFENZUSTELLUNG
23 APPR LCT X+10 Y+10 R5 RL F200 ..... KONTUR-STARTPUNKT
                                         WEICHES ANFAHREN

24 L Y+45
25 L X+30 Y+10
26 L X+10 Y+10
27 DEP LCT X+0 Y+0 R5 ..... WEICHES WEGFAHREN
28 LBL 0
29 END PGM 229 MM
    
```







HAUPTPROGRAMM

```

0 BEGIN PGM 232 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S4000 ..... R3
4 L Z+100 R0 F MAX M3
5 CYCL DEF 7.0 NULLPUNKT
6 CYCL DEF 7.1 X+20
7 CYCL DEF 7.2 Y+10
8 CALL LBL 1

9 CYCL DEF 7.0 NULLPUNKT
10 CYCL DEF 7.1 X+60
11 CYCL DEF 7.2 Y+10
12 CYCL DEF 11.0 MASSFAKTOR
13 CYCL DEF 11.1 SCL1,2
14 CALL LBL 1

15 CYCL DEF 11.0 MASSFAKTOR
16 CYCL DEF 11.1 SCL1
17 CYCL DEF 7.0 NULLPUNKT
18 CYCL DEF 7.1 X+20
19 CYCL DEF 7.2 Y+60
20 CYCL DEF 10.0 DREHUNG
21 CYCL DEF 10.1 ROT+20
22 CALL LBL 1

23 CYCL DEF 10.0 DREHUNG
24 CYCL DEF 10.1 ROT+0
25 CYCL DEF 7.0 NULLPUNKT
26 CYCL DEF 7.1 X+90
27 CYCL DEF 7.2 Y+60
28 CYCL DEF 8.0 SPIEGELN
29 CYCL DEF 8.1 X
30 CALL LBL 1
    
```



31 CYCL DEF 8.0 SPIEGELN
32 CYCL DEF 8.1
33 CYCL DEF 7.0 NULLPUNKT
34 CYCL DEF 7.1 X+0
35 CYCL DEF 7.2 Y+0



Freifahren, Ende


36 L Z+100 R0 F MAX M2

UP, Kontur

37 LBL 1
38 L X+10 Y+10 R0 F MAX M3
39 L Z+2 F MAX
40 L Z-15 R0 F100
41 APPR LCT X+0 Y+10 R2 RR
42 L Y+20 X+0
43 RND R4
44 L X+20 Y+20
45 RND R4
46 L Y+0
47 CC X+15 Y+0
49 C X+10 Y+0 DR-
49 L X+0 Y+0
50 RND R4
51 L X+0 Y+10
52 DEP LCT X+10 Y+10 R2
53 L Z+2 R0 F MAX
54 LBL 0
55 END PGM 232 MM







HAUPTPROGRAMM

```

0 BEGIN PGM 284 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-10
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S2000..... R3
4 CYCL DEF 7.0 NULLPUNKT
5 CYCL DEF 7.1 X+5
6 CYCL DEF 7.2 Y+5
7 CALL LBL 1

8 CYCL DEF 7.0 NULLPUNKT
9 CYCL DEF 7.1 X+95
10 CYCL DEF 7.2 Y+15
11 CYCL DEF 10.0 DREHUNG
12 CYCL DEF 10.1 ROT+105
13 CALL LBL 1

14 CYCL DEF 10.0 DREHUNG
15 CYCL DEF 10.1 ROT+0
16 CYCL DEF 7.0 NULLPUNKT
17 CYCL DEF 7.1 X+20
18 CYCL DEF 7.2 Y+95
19 CYCL DEF 8.0 SPIEGELN
20 CYCL DEF 8.1 Y
21 CYCL DEF 26.0 MASSFAKTOR ACHSSP.
22 CYCL DEF 26.1 X1,9 Y1 CCX+10 CCY+95
23 CALL LBL 1

24 CYCL DEF 26.0 MASSFAKTOR ACHSSP.
25 CYCL DEF 26.1 X1 Y1 CCX+0 CCY+0
26 CYCL DEF 8.0 SPIEGELN
27 CYCL DEF 8.1
28 CYCL DEF 7.0 NULLPUNKT
29 CYCL DEF 7.1 X+0
30 CYCL DEF 7.2 Y+0

31 L Z+100 R0 F MAX M2
    
```





Lösung:


Maßfaktor – Hammer

UP

```
32 LBL 1
33 L X+7 Y+15 R0 F300
34 L Z-5
35 APPR LCT X+0 Y+15 R1 RR
36 L Y+25
37 L X+5 Y+40
38 L X+10 Y+25
39 L Y+20
40 L X+40
41 L Y+10
42 L X+10
43 L Y+0
44 L X+0
45 L Y+15
46 DEP LCT X+8 Y+15 R1
47 L Z+2 R0
48 LBL 0
49 END PGM 284 MM
```







HAUPTPROGRAMM

```

0 BEGIN PGM 234 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX M3
5 CALL LBL 1 ..... UP-AUFRUF TEIL ①

6 CYCL DEF 7.0 NULLPUNKT
7 CYCL DEF 7.1 X+50
8 CYCL DEF 7.2 Y+50
9 CYCL DEF 26.0 MASSFAKTOR ACHSSP.
10 CYCL DEF 26.1 X1,5 CCX+10
11 CALL LBL 1 ..... UP-AUFRUF TEIL ②

12 CYCL DEF 7.0 NULLPUNKT
13 CYCL DEF 7.1 X+0
14 CYCL DEF 7.2 Y+0
15 CYCL DEF 26.0 MASSFAKTOR ACHSSP.
16 CYCL DEF 26.1 X1 ..... ZYKLUS RUECKSETZEN



Freifahren, Ende 17 L Z+100 R0 F MAX M2
    
```


UP, Kontur

```

18 LBL 1
19 L X+0 Y+0 R0 F MAX
20 L Z+2 F MAX
21 L Z-10 R0 F100 M8
22 APPR LT X+10 Y+10 LEN5 RL F200
23 L IY+20
24 L IX+21
25 L IX+4 IY-4
26 L Y+10
27 L X+10
28 DEP LT LEN5
29 L X+0 Y+0 R0
30 L Z+2 R0 F MAX
31 LBL 0
32 END PGM 234 MM
    
```







Gesamtprogramm



```


0 BEGIN PGM 285 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 19 Z S4000 ..... WERKZEUG RUFEN, L ABSOLUT
                               IN WERKZEUG-TABELLE!
4 CYCL DEF 1.0 TIEFBOHREN ..... ZYKLUS DEFINIEREN
5 CYCL DEF 1.1 ABST +2
6 CYCL DEF 1.2 TIEFE -20
7 CYCL DEF 1.3 ZUSTLG -5
8 CYCL DEF 1.4 V.ZEIT 0
9 CYCL DEF 1.5 F400

10 L Z+100 R0 F MAX ..... WERKZEUG-ACHSE FREIFAHREN
11 L X+100 Y+50 R0 F MAX ..... VORPOSITIONIEREN
12 L B+27 R0 F MAX ..... KOPF SCHWENKEN (FAHREN)

13 CYCL DEF 19.0 BEARBEITUNGSEBENE
14 CYCL DEF 19.1 B+27 ..... KOPF SCHWENKEN (RECHNEN)
15 L X+50 Y+50 R0 F MAX M3 ..... BOHRPOSITION ANFAHREN
16 L Z+2 R0 F MAX M99 ..... Z-SICHERHEIT, BOHRZ. RUFEN
17 CYCL DEF 19.0 BEARBEITUNGSEBENE
18 CYCL DEF 19.1 B+0 ..... SCHWENKEN RUECKSETZEN

19 L Z+100 R0 F MAX M5 ..... WKZ FREIFAHREN
20 L B+0 R0 F MAX M30 ..... KOPF ZURUECK SCHWENKEN
21 END PGM 285 MM
    
```





Gesamtprogramm

```

0 BEGIN PGM 286 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-40
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 19 Z S4000 ..... WERKZEUG RUFEN, L ABSOLUT
                               IN WERKZEUG-TABELLE!
4 CYCL DEF 1.0 TIEFBOHREN ..... ZYKLUS DEFINIEREN
5 CYCL DEF 1.1 ABST +2
6 CYCL DEF 1.2 TIEFE -20
7 CYCL DEF 1.3 ZUSTLG -5
8 CYCL DEF 1.4 V.ZEIT 0
9 CYCL DEF 1.5 F400
10 L Z+100 R0 F MAX ..... WERKZEUG-ACHSE FREIFAHREN
11 L X+150 R0 F MAX ..... VORPOSITIONIEREN
12 CYCL DEF 7.0 NULLPUNKT
13 CYCL DEF 7.1 X+50 ..... NULLPUNKT AUF RECHTE KANTE
14 L B+90 R0 F MAX ..... KOPF SCHWENKEN (FAHREN)

15 CYCL DEF 19.0 BEARBEITUNGSEBENE
16 CYCL DEF 19.1 B+90 ..... KOPF SCHWENKEN (RECHNEN)
17 L X+25 Y+0 R0 F MAX M3 ..... BOHRPOSITION ANFAHREN
18 L Z+2 R0 F MAX M99 ..... Z-SICHERHEIT, BOHRZ. RUFEN
19 CYCL DEF 19.0 BEARBEITUNGSEBENE
20 CYCL DEF 19.1 B+0 ..... SCHWENKEN RUECKSETZEN

21 CYCL DEF 7.0 NULLPUNKT
22 CYCL DEF 7.1 X+0
23 L Z+100 R0 F MAX M5 ..... WKZ FREIFAHREN
24 L B+0 R0 F MAX M30 ..... KOPF ZURUECK SCHWENKEN
25 END PGM 286 MM
    
```

